Introduction to Symbolic Computations in GR with xTensor

Downloading and Installing xAct

Visit the page: http://www.xact.es
Follow installation instructions on http://www.xact.es/download.html

Linux:

1. Download the tarball xAct_V.tgz (V is the version number)
2. sudo -i ; cd /usr/share/Mathematica/Applications/; tar xvfz ~/Downloads/xAct_V.tgz

Windows:

1. Download the zip file xAct_V.zip (V is the version number)
2. unzip its contents in C:\Program Files\Wolfram Research\Mathematica\<version>\AddOns\Applications\

<u>Read the documentation:</u>

http://www.xact.es/documentation.html
If you want to use xTensor, you will not avoid reading the full documentation. Better earlier than later: xTensorDoc.nb
The reference notebook is useful too: xTensorRefGuide.nb

The documentation is also installed locally, most likely in:
<u>Linux:</u> /usr/share/Mathematica/Applications/xAct/Documentation/English/
<u>Windows:</u> C:\Program Files\Wolfram Research\Mathematica\<version>\AddOns\Applications\xAct\Documentation\English

Explore the documentation in xTensorDoc.nb.... Make a copy to the notebook, so that you can play with it.

```
cp /usr/share/Mathematica/Applications/xAct/Documentation/English/xTensorDoc.nb .
cp /usr/share/Mathematica/Applications/xAct/Documentation/English/xTensorRefGuide.nb .
```

Start an xTensor session

If you are already running a Mathematica session, esp. if you have loaded xTerior, xAct, ..., make a call to Quit[] before the loading of the package:
Quit[]; Needs[“xAct`xTensor`”]

In[ ]:=
```
Needs["xAct`xTensor`"]
```

```
------------------------------------------------------------
Package xAct`xPerm`  version 1.2.3, {2015, 8, 23}
CopyRight (C) 2003-2020, Jose M. Martin-Garcia, under the General Public License.
Connecting to external linux executable...
Connection established.
------------------------------------------------------------
Package xAct`xTensor`  version 1.2.0, {2021, 10, 17}
CopyRight (C) 2002-2021, Jose M. Martin-Garcia, under the General Public License.
------------------------------------------------------------
These packages come with ABSOLUTELY NO WARRANTY; for details type
  Disclaimer[]. This is free software, and you are welcome to redistribute
  it under certain conditions. See the General Public License for details.
------------------------------------------------------------
```

Define a manifold, whose name is M4, and its dimension is 4.
The indices used are {λ,μ,ν,ρ,σ,α,β,γ,δ}.You can put any indices you like, they should consist of one character.

Online documentation:

In[ ]:=
```
? DefManifold
```

Out[ ]=

Symbol

DefManifold[M, dim, {a, b, c,...}] defines M to be an n-dimensional differentiable manifold with dimension dim (a positive integer or a constant symbol) and tensor abstract indices a, b, c, ... DefManifold[M, {M1, ..., Mm}, {a, b, c,...}] defines M to be the product manifold of previously defined manifolds M1 ... Mm. For backward compatibility dim can be a list of positive integers, whose length is interpreted as the dimension of the defined manifold.

See all available functions.
Click on any name for usage information:

In[ ]:=
```
?xAct`xTensor`*
```

In[ ]:=
```
DefManifold[M4, 4, {λ, μ, ν, ρ, σ, α, β, γ, δ}];
```

```
** DefManifold: Defining manifold M4.

** DefVBundle: Defining vbundle TangentM4.
```

DimOfManifold[M4] gives its dimension. A tangent bundle is associated with the manifold, you can refer to it as TangentM4
This is the first indication that xAct defines objects by creating Heads that have names related to the definition of the names of the Manifold, the metric, the covariant derivative, etc.

In[ ]:=
```
Print[
 "The Manifold M4 has dimension ", DimOfManifold[M4], "\n",
 "Its tangent bundle is ", Tangent[M4],
 " and you can refer to is using the name ",
 Tangent[M4] // InputForm, "   (Tangent + M4 = TangentM4)\n",
 "For example: The indices of ",
 TangentM4, " are: ", IndicesOfVBundle[TangentM4]
]
```

```
The Manifold M4 has dimension 4
Its tangent bundle is TM4
  and you can refer to is using the name TangentM4   (Tangent + M4 = TangentM4)
For example: The indices of TM4 are: {{λ, μ, ν, ρ, σ, α, β, γ, δ}, {}}
```

We define a metric, whose symbol will be “g” and a metric compatible, torsion free covariant derivative CD. A lot of things are defined with it by default:
1. A Christoffel covariant derivative (compatible with the metric+torsion free) with name CD and its Christoffel symbols ChristoffelCD.
2. A Levi-Civita tensor $\epsilon_{\alpha\beta\gamma\delta}$ epsilong[ ] and the metric determinant Detg[ ]
3. Curvature tensors RiemannCD, RicciCD, RiemannScalarCD[ ], EinsteinCD[ ], WeylCD[ ], ...

Read the output messages printed after evaluating the cell below for more information.

Notice that the names of the symbols referring to the above tensors/tensor densities end in “g” or “CD”. This comes from the chosen (by default here) names. If the metric had name, e.g. **MetricG**, and the covariant derivative **CDer**, then we would have epsilon**MetricG**, Det**MetricG**, Riemann**CDer**, Ricci**CDer**, etc..

The -1 in the first argument, means that the determinant of the metric is negative

In[ ]:=
```
DefMetric[-1, g[-μ, -ν], CD];
```

```
** DefTensor: Defining symmetric metric tensor g[-μ, -ν].
** DefTensor: Defining antisymmetric tensor epsilong[-α, -β, -γ, -δ].
** DefTensor: Defining tetrametric Tetrag[-α, -β, -γ, -δ].
** DefTensor: Defining tetrametric Tetrag†[-α, -β, -γ, -δ].
** DefCovD: Defining covariant derivative CD[-μ].
** DefTensor: Defining vanishing torsion tensor TorsionCD[α, -β, -γ].
** DefTensor: Defining symmetric Christoffel tensor ChristoffelCD[α, -β, -γ].
** DefTensor: Defining Riemann tensor RiemannCD[-α, -β, -γ, -δ].
** DefTensor: Defining symmetric Ricci tensor RicciCD[-α, -β].
** DefCovD:  Contractions of Riemann automatically replaced by Ricci.
** DefTensor: Defining Ricci scalar RicciScalarCD[].
** DefCovD:  Contractions of Ricci automatically replaced by RicciScalar.
** DefTensor: Defining symmetric Einstein tensor EinsteinCD[-α, -β].
** DefTensor: Defining Weyl tensor WeylCD[-α, -β, -γ, -δ].
** DefTensor: Defining symmetric TFRicci tensor TFRicciCD[-α, -β].
** DefTensor: Defining Kretschmann scalar KretschmannCD[].
** DefCovD:  Computing RiemannToWeylRules for dim 4
** DefCovD:  Computing RicciToTFRicci for dim 4
** DefCovD:  Computing RicciToEinsteinRules for dim 4
** DefTensor: Defining weight +2 density Detg[]. Determinant.
```

Tensors are symbolic objects that take indices. An index that has a minus sign in front of it is contravariant, otherwise it is covariant.

Tensors must be defined using DefTensor: The indices must be among the list of indices in the tangent bundle of M4, TangentM4.

```
DefTensor[v[ α], M4];
DefTensor[u[ α], M4]; (* vectors: covariant,     (1,0) tensors *)
DefTensor[ω[-α], M4];
DefTensor[ξ[-α], M4]; (* 1-forms: contravariant, (0,1) tensors *)
{v[μ], u[ν], ω[-β], ξ[-γ]}
```

```
** DefTensor: Defining tensor v[α].
** DefTensor: Defining tensor u[α].
** DefTensor: Defining tensor ω[-α].
** DefTensor: Defining tensor ξ[-α].
```

Out[ ]= $\{v^{\mu}, u^{\mu}, \omega_{\mu}, \xi_{\mu}\}$

Here are more tensors. We can specify if (some of) their indices have a specific symmetry:

```
DefTensor[F[-α, -β]          , M4, Antisymmetric[{-α, -β}]];
(* (0,2) tensor, antisymmetric in α,β *)
DefTensor[S[-α, -β]          , M4,    Symmetric[{-α, -β}]];
(* (0,2) tensor,     symmetric in α,β *)
{F[-μ, -ν], S[-α, -β]}
```

```
** DefTensor: Defining tensor F[-α, -β].
** DefTensor: Defining tensor S[-α, -β].
```

In[ ]:=
```
DefTensor[Stensor[ α, -β, -γ], M4,    Symmetric[{-β, -γ}], PrintAs → "𝒮"];
(* (1,1) tensor,symmetric in β,γ  note: [Esc]scS[Esc] gives 𝒮 *)
Stensor[μ, -ν, -λ]
```

```
** DefTensor: Defining tensor Stensor[α, -β, -γ].
```

Out[ ]= $\mathcal{S}^{\mu}{}_{\nu\lambda}$

If you don’t need a tensor anymore, or you need to redefine it, you have to undefine it first, using UndefTensor:

In[ ]:=
```
UndefTensor[Stensor]
```

```
** UndefTensor: Undefined tensor Stensor
```

ϕ is a scalar field; c a *constant* scalar field: it is defined on no manifold and its derivative is zero (see below). We also define a constant.

In[ ]:=
```
DefTensor[scalar[], M4, PrintAs → "ϕ"];
DefTensor[const[], {}, PrintAs → "c"] ; DefConstantSymbol[hbar, PrintAs → "ℏ"];
{scalar[], const[], hbar}
```

```
** DefTensor: Defining tensor scalar[].
** DefTensor: Defining tensor const[].
** DefConstantSymbol: Defining constant symbol hbar.
```

Out[ ]= $\{\phi, c, \hbar\}$

If you get stuck, or make a serious error, you may need to restart your computation. First call Quit[] and then reevaluate the notebook: Evaluation → Evaluate Notebook

In[ ]:=
```
(* Quit *)
```

Contractions: repeated upstairs+downstairs indices are contracted

In[∘]:=
```
{v[μ] ω[-μ], v[μ] v[-μ],
 S[-α, -β] u[α],
 F[-μ, -ν] v[μ] u[ν], F[-μ, -ν] v[μ] v[ν], F[-μ, -ν] v[μ] v[ν] // ToCanonical,
 F[-μ, -ν] S[μ, ν], F[-μ, -ν] S[μ, ν] // ToCanonical}
```

Out[∘]= $\{v^{\mu}\ \omega_{\mu},\ v_{\mu}\ v^{\mu},\ S_{\alpha\beta}\ u^{\alpha},\ F_{\mu\nu}\ u^{\nu}\ v^{\mu},\ F_{\mu\nu}\ v^{\mu}\ v^{\nu},\ 0,\ F_{\mu\nu}\ S^{\mu\nu},\ 0\}$

If a tensorial expression is not valid - in the sense of the abstract index notation - Validate captures it:

In[∘]:=
```
{v[μ] v[ν] // Validate, v[μ] v[μ] // Validate, v[-μ] v[-μ] // Validate}
```

Validate: Found indices with the same name μ.

Validate: Found indices with the same name −μ.

Out[∘]= $\{v^{\mu}\ v^{\nu},\ \mathrm{Null},\ \mathrm{Null}\}$

Let’s write down expressions using the tensors constructed for the metric g: we use the abstract index notation (see Wald)
Due to the metric, we can write ξ_{λ}, as well as ξ^{λ}.

In[∘]:=
```
RiemannCD[-μ, -ν, -λ, σ] v[μ] u[ν] ξ[λ] ω[-σ]
```

Out[∘]= $R[\nabla]_{\mu\nu\lambda}{}^{\sigma}\ u^{\nu}\ v^{\mu}\ \xi^{\lambda}\ \omega_{\sigma}$

xTensor knows that the Riemann tensor is antisymmetric, need to use ToCanonical to apply it:

In[∘]:=
```
{RiemannCD[-μ, -ν, -λ, σ] v[μ] v[ν] ξ[λ] ω[-σ],
 RiemannCD[-μ, -ν, -λ, σ] v[μ] v[ν] ξ[λ] ω[-σ] // ToCanonical}
```

Out[∘]= $\{R[\nabla]_{\mu\nu\lambda}{}^{\sigma}\ v^{\mu}\ v^{\nu}\ \xi^{\lambda}\ \omega_{\sigma},\ 0\}$

The *monoterm* symmetries of $R_{\mu\nu\rho\sigma}$ are also known: (unfortunately “there is no algorithm at present in `xTensor`` to canonicalize expressions with *multiterm* symmetries” - section 9.2 of xTensorDoc.nb)

In[∘]:=
```
{RiemannCD[-μ, -ν, -λ, -σ] v[μ] u[ν] ξ[λ] ξ[σ],
 RiemannCD[-μ, -ν, -λ, -σ] v[μ] u[ν] ξ[λ] ξ[σ] // ToCanonical}
```

Out[∘]= $\{R[\nabla]_{\mu\nu\lambda\sigma}\ u^{\nu}\ v^{\mu}\ \xi^{\lambda}\ \xi^{\sigma},\ 0\}$

$$R_{\mu\nu\lambda\sigma}\, F^{\mu\lambda}\, u^{\nu}\, u^{\sigma} = R_{\lambda\sigma\,\mu\nu}\, F^{\mu\lambda}\, u^{\nu}\, u^{\sigma} = -R_{\lambda\sigma\,\mu\nu}\, F^{\lambda\mu}\, u^{\nu}\, u^{\sigma} = -R_{\mu\sigma\,\lambda\nu}\, F^{\mu\lambda}\, u^{\nu}\, u^{\sigma} = -R_{\mu\nu\,\lambda\sigma}\, F^{\mu\lambda}\, u^{\sigma}\, u^{\nu} = 0$$

In[∘]:=
```
{RiemannCD[-μ, -ν, -λ, -σ] F[μ, λ] u[σ] u[ν],
 RiemannCD[-μ, -ν, -λ, -σ] F[μ, λ] u[σ] u[ν] // ToCanonical}
```

Out[∘]= $\{F^{\mu\lambda}\ R[\nabla]_{\mu\nu\lambda\sigma}\ u^{\nu}\ u^{\sigma},\ 0\}$

In[ ]:=
```
{RiemannCD[-μ, -ν, -λ, -σ] F[μ, λ] RicciCD[σ, ν],
 RiemannCD[-μ, -ν, -λ, -σ] F[μ, λ] RicciCD[σ, ν] // ToCanonical}
```

Out[ ]= $\{F^{\mu\lambda}\ R[\nabla]^{\sigma\nu}\ R[\nabla]_{\mu\nu\lambda\sigma},\ 0\}$

The Kronecker's deltas

The Kronecker's deltas are defined:

$\delta_\mu{}^\nu,\ \delta_{\mu 1\mu 2\ldots\mu n}{}^{\nu 1\nu 2\ldots\nu n} = n!\,\delta_{[\mu 1}{}^{\nu 1}\,\delta_{\mu 2}{}^{\nu 2}\ldots\delta_{\mu n]}{}^{\nu n}$

ExpandGdelta[Gdelta[...]] expands in terms of delta[ ]. If there is a metric defined, we can have all indices up or down: indices are contracted using the metric.

In[ ]:=
```
Print[
 "\n--------------------------------------------------------------",
 "\n δ_μ^ν =  ", delta[-μ, ν], ",   δ_μ^μ = ",
 delta[-μ, μ], ",   δ_μ^ν ω_ν = ", delta[-μ, ν] ω[-ν],
 "\n δ_μν^ρσ= ", Gdelta[-μ, -ν, ρ, σ], " = ", ExpandGdelta[Gdelta[-μ, -ν, ρ, σ]],
 "  δ_μν^ρσ ω_ρ v^ν =  ", Gdelta[-μ, -ν, ρ, σ] ω[-ρ] v[ν],
 "\n δ_μνρ^αβγ=", ExpandGdelta[Gdelta[-μ, -ν, -ρ,  α,  β,  γ]],
 "\n δ_μνραβγ=", ExpandGdelta[Gdelta[-μ, -ν, -ρ, -α, -β, -γ]],
 "  (valid only if metric defined)"
]
```

$\delta_\mu{}^\nu = \delta_\mu{}^\nu,\quad \delta_\mu{}^\mu = 4,\quad \delta_\mu{}^\nu\ \omega_\nu = \omega_\mu$

$\delta_{\mu\nu}{}^{\rho\sigma} = \delta_{\mu\nu}{}^{\rho\sigma} = -\delta_\mu{}^\sigma\ \delta_\nu{}^\rho + \delta_\mu{}^\rho\ \delta_\nu{}^\sigma\quad \delta_{\mu\nu}{}^{\rho\sigma}\ \omega_\rho\ v^\nu = v^\sigma\ \omega_\mu - \delta_\mu{}^\sigma\ v^\rho\ \omega_\rho$

$\delta_{\mu\nu\rho}{}^{\alpha\beta\gamma} =$

$-\delta_\mu{}^\gamma\ \delta_\nu{}^\beta\ \delta_\rho{}^\alpha + \delta_\mu{}^\beta\ \delta_\nu{}^\gamma\ \delta_\rho{}^\alpha + \delta_\mu{}^\gamma\ \delta_\nu{}^\alpha\ \delta_\rho{}^\beta - \delta_\mu{}^\alpha\ \delta_\nu{}^\gamma\ \delta_\rho{}^\beta - \delta_\mu{}^\beta\ \delta_\nu{}^\alpha\ \delta_\rho{}^\gamma + \delta_\mu{}^\alpha\ \delta_\nu{}^\beta\ \delta_\rho{}^\gamma$

$\delta_{\mu\nu\rho\alpha\beta\gamma} = -g_{\mu\gamma}\ g_{\nu\beta}\ g_{\rho\alpha} + g_{\mu\beta}\ g_{\nu\gamma}\ g_{\rho\alpha} + g_{\mu\gamma}\ g_{\nu\alpha}\ g_{\rho\beta} - g_{\mu\alpha}\ g_{\nu\gamma}\ g_{\rho\beta} - g_{\mu\beta}\ g_{\nu\alpha}\ g_{\rho\gamma} + g_{\mu\alpha}\ g_{\nu\beta}\ g_{\rho\gamma}$

(valid only if metric defined)

Levi-Civita tensor

When a metric is defined, the Levi Civita tensor $\epsilon_{\mu 1\mu 2\ldots\mu n}$ is defined, and we can do contractions:

In[ ]:=
```
Print[
 "------------------------------------------------------------\n",
 epsilong[-α, -β, -γ, -δ], " ,     ", epsilong[-α, -β, -γ, -δ] g[α, β],
 " ,     ", epsilong[-α, -β, -γ, -δ] g[α, β] // ToCanonical, "\n",
 "------------------------------------------------------------\n",
 epsilong[-α, -β, -γ, -δ] epsilong[-μ, -ν, -ρ, δ], "\n",
 "------------------------------------------------------------\n",
 epsilong[-α, -β, -γ, -δ] epsilong[μ, ν, ρ, δ], "\n",
 epsilong[-α, -β, -γ, -δ] epsilong[μ, ν, γ, δ], "\n",
 epsilong[-α, -β, -γ, -δ] epsilong[μ, β, γ, δ], "\n",
 epsilong[-α, -β, -γ, -δ] epsilong[α, β, γ, δ], " = 4!\n",
 epsilong[-α, -β, -γ, -δ] epsilong[μ, ν, ρ, δ] epsilong[ α, β, γ, λ]
]
```

$\epsilon g_{\alpha\beta\gamma\delta}$, $\epsilon g_{\alpha\beta\gamma\delta}\ g^{\alpha\beta}$,   0

$g_{\alpha\rho}\ g_{\beta\nu}\ g_{\gamma\mu} - g_{\alpha\nu}\ g_{\beta\rho}\ g_{\gamma\mu} - g_{\alpha\rho}\ g_{\beta\mu}\ g_{\gamma\nu} + g_{\alpha\mu}\ g_{\beta\rho}\ g_{\gamma\nu} + g_{\alpha\nu}\ g_{\beta\mu}\ g_{\gamma\rho} - g_{\alpha\mu}\ g_{\beta\nu}\ g_{\gamma\rho}$

$\delta_\alpha{}^\rho\ \delta_\beta{}^\nu\ \delta_\gamma{}^\mu - \delta_\alpha{}^\nu\ \delta_\beta{}^\rho\ \delta_\gamma{}^\mu - \delta_\alpha{}^\rho\ \delta_\beta{}^\mu\ \delta_\gamma{}^\nu + \delta_\alpha{}^\mu\ \delta_\beta{}^\rho\ \delta_\gamma{}^\nu + \delta_\alpha{}^\nu\ \delta_\beta{}^\mu\ \delta_\gamma{}^\rho - \delta_\alpha{}^\mu\ \delta_\beta{}^\nu\ \delta_\gamma{}^\rho$

$2\ \delta_\alpha{}^\nu\ \delta_\beta{}^\mu - 2\ \delta_\alpha{}^\mu\ \delta_\beta{}^\nu$

$-6\ \delta_\alpha{}^\mu$

-24 = 4!

$-6\ \epsilon g^{\mu\nu\rho\lambda}$

Metric Contractions

Use ContractMetric to contract indices:

In[ ]:=
```
Print[
 "\n-------------------------------------------------------------",
 "\n gμσg^σν     =  ", g[-μ, -σ] g[σ, ν]            ,
 "    gμ^μ  = ", g[-μ, μ], "     gμνg^μν  = ", g[-μ, -ν] g[μ, ν],
 "\n-------------------------------------------------------------",
 "\n gμνv^ν     =  ", g[-μ, -ν]     v[ν]            ,
 " = ", g[-μ, -ν]      v[ ν]           // ContractMetric,
 "\n g^μνων     =  ", g[μ, ν]          ω[-ν]          ,
 " = ", g[μ, ν]            ω[-ν]           // ContractMetric,
 "\n g^μνFνσ    =  ", g[μ, ν]         F[-ν, -σ]       ,
 " = ", g[μ, ν]          F[-ν, -σ]       // ContractMetric,
 "\n g^νσFμνωσ  =  ", g[ν, σ]         F[-μ, -ν] ω[-σ],
 " = ", g[ν, σ]          F[-μ, -ν] ω[-σ]  // ContractMetric,
 "\n g^μνg^ρσFνσ =  ", g[μ, ν] g[ρ, σ] F[-ν, -σ]       ,
 " = ", g[μ, ν] g[ρ, σ] F[-ν, -σ]       // ContractMetric,
 "\n",  S[-μ, -ν] g[μ, ν], "  =   ", S[-μ, -ν] g[μ, ν] // ContractMetric,
 "\n-------------------------------------------------------------",
 "\n",  RiemannCD[-μ, -ν, -ρ, -σ] g[ν, ρ] , "  =  ",
 RiemannCD[-μ, -ν, -ρ, -σ] g[ν, ρ] // ContractMetric,
 "\n", "Rμνg^μν  = ", RicciCD[-μ, -ν] g[μ, ν], "  = ",
 RicciCD[-μ, -ν] g[μ, ν] // ContractMetric, "    (no need to use ContractMetric)"
]
```

$g_{\mu\sigma}g^{\sigma\nu} = \delta_\mu{}^\nu \quad g_\mu{}^\mu = 4 \quad g_{\mu\nu}g^{\mu\nu} = 4$

$g_{\mu\nu}v^\nu = g_{\mu\nu}\; v^\nu = v_\mu$

$g^{\mu\nu}\omega_\nu = g^{\mu\nu}\; \omega_\nu = \omega^\mu$

$g^{\mu\nu}F_{\nu\sigma} = F_{\nu\sigma}\; g^{\mu\nu} = F^\mu{}_\sigma$

$g^{\nu\sigma}F_{\mu\nu}\omega_\sigma = F_{\mu\nu}\; g^{\nu\sigma}\; \omega_\sigma = F_{\mu\nu}\; \omega^\nu$

$g^{\mu\nu}g^{\rho\sigma}F_{\nu\sigma} = F_{\nu\sigma}\; g^{\mu\nu}\; g^{\rho\sigma} = F^{\mu\rho}$

$g^{\mu\nu}\; S_{\mu\nu} = S_\mu{}^\mu$

$g^{\nu\rho}\; R[\nabla]_{\mu\nu\rho\sigma} = -R[\nabla]_{\mu\sigma}$

$R_{\mu\nu}g^{\mu\nu} = R[\nabla] = R[\nabla]$ (no need to use ContractMetric)

Covariant Derivative

We define a scalar field f[], a *constant* scalar field ϕ (it is defined on no manifold), and a constant.

In[ ]:=
```
DefTensor[f[], M4]; DefTensor[ϕ[], {}] ; DefConstantSymbol[GNewton, PrintAs → "GN"];
```

** DefTensor: Defining tensor f[].

** DefTensor: Defining tensor ϕ[].

** DefConstantSymbol: Defining constant symbol GNewton.

The predefined, ordinary, flat, partial derivative: $[\partial_\mu, \partial_\nu] = 0$

In[ ]:=
```
Print[
  PD[-μ][f[]], "\n",
  PD[-μ][v[ν]], "\n",
  PD[-μ][f[] v[ν]], "\n",
  PD[-μ][u[ρ] v[ν]], "\n",
  PD[-μ][PD[-ν][u[σ]]] - PD[-ν][PD[-μ][u[σ]]], " = ",
  PD[-μ][PD[-ν][u[σ]]] - PD[-ν][PD[-μ][u[σ]]] // ToCanonical
]
```

$\partial_\mu f$

$\partial_\mu v^\nu$

$v^\nu \partial_\mu f + f \partial_\mu v^\nu$

$v^\nu \partial_\mu u^\rho + u^\rho \partial_\mu v^\nu$

$\partial_\mu \partial_\nu u^\sigma - \partial_\nu \partial_\mu u^\sigma = 0$

The Covariant derivative:

In[ ]:=
```
Print[
 "∇μ GN = "                  ,
 CD[-μ] [GNewton]                                        , "\n",
 CD[-μ] [f[]]        , " = ",
 CD[-μ] [f[]]        // CovDToChristoffel                , "\n",
 CD[-μ] [GNewton f[]], " = ",
 CD[-μ] [GNewton f[]] // CovDToChristoffel               , "\n",
 CD[-μ][ϕ[]]         , " = ",
 CD[-μ][ϕ[]]         // CheckZeroDerivative              ,
 "  (must use CheckZeroDerivative to
   actually see that the derivative is zero)\n",
 CD[-μ][v[ν]]        , " = ",
 CD[-μ][v[ν]]        // CovDToChristoffel // ScreenDollarIndices, "\n",
 CD[-μ][ω[-ν]]       , " = ",
 CD[-μ][ω[-ν]]       // CovDToChristoffel // ScreenDollarIndices, "\n",
 CD[-μ][F[-ν, -ρ]]   , " = ",
 CD[-μ][F[-ν, -ρ]]   // CovDToChristoffel // ScreenDollarIndices, "\n",
 CD[-μ][f[] v[ν]]   , " = ",
 CD[-μ][f[] v[ν]]   // CovDToChristoffel // ScreenDollarIndices, "\n",
 "∇μ (Fνρv^ν) = "            , CD[-μ][F[-ν, -ρ] v[ν]]
]
```

$\nabla_\mu\ G_N = 0$

$\nabla_\mu f = \partial_\mu f$

$G_N(\nabla_\mu f) = G_N\,\partial_\mu f$

$\nabla_\mu \phi = 0$ (must use CheckZeroDerivative to actually see that the derivative is zero)

$\nabla_\mu v^\nu = \Gamma[\nabla]^\nu{}_{\mu\alpha}\ v^\alpha + \partial_\mu v^\nu$

$\nabla_\mu \omega_\nu = -\Gamma[\nabla]^\alpha{}_{\mu\nu}\ \omega_\alpha + \partial_\mu \omega_\nu$

$\nabla_\mu F_{\nu\rho} = -\Gamma[\nabla]^\alpha{}_{\mu\nu}\ F_{\alpha\rho} - \Gamma[\nabla]^\alpha{}_{\mu\rho}\ F_{\nu\alpha} + \partial_\mu F_{\nu\rho}$

$v^\nu\ (\nabla_\mu f) + f(\nabla_\mu v^\nu) = v^\nu\ \partial_\mu f + f\left(\Gamma[\nabla]^\nu{}_{\mu\alpha}\ v^\alpha + \partial_\mu v^\nu\right)$

$\nabla_\mu\ (F_{\nu\rho}v^\nu) = v^\nu\ (\nabla_\mu F_{\nu\rho}) + F_{\nu\rho}\ (\nabla_\mu v^\nu)$

In[ ]:=
```
Print[
 ChristoffelCD[ μ, -ν, -ρ], " = ",
 ChristoffelCD[ μ, -ν, -ρ] // ChristoffelToGradMetric // ScreenDollarIndices, "\n",
 ChristoffelCD[-μ, -ν, -ρ], " = ",
 ChristoffelCD[-μ, -ν, -ρ] // ChristoffelToGradMetric // ScreenDollarIndices, "\n"
]
```

$\Gamma[\nabla]^{\mu}{}_{\nu\rho} \;=\; \frac{1}{2}\; g^{\mu\alpha}\left(-\partial_{\alpha} g_{\nu\rho} + \partial_{\nu} g_{\rho\alpha} + \partial_{\rho} g_{\nu\alpha}\right)$

$\Gamma[\nabla]_{\mu\nu\rho} \;=\; \frac{1}{2}\left(-\partial_{\mu} g_{\nu\rho} + \partial_{\nu} g_{\rho\mu} + \partial_{\rho} g_{\nu\mu}\right)$

In[ ]:=
```
Print[
 CD[-μ][v[ν]]                                                          ,
 "= ",
 CD[-μ][v[ν]]  // CovDToChristoffel //
  ScreenDollarIndices                              , "\n = ",
 CD[-μ][v[ν]]  // CovDToChristoffel // ChristoffelToGradMetric //
  ScreenDollarIndices, "\n = ",
 CD[-μ][v[ν]]  // CovDToChristoffel // ChristoffelToGradMetric // ToCanonical //
  ScreenDollarIndices, "\n"
]
```

$\nabla_{\mu} v^{\nu} = \;\Gamma[\nabla]^{\nu}{}_{\mu\alpha}\; v^{\alpha} + \partial_{\mu} v^{\nu}$

$= \frac{1}{2}\; g^{\nu\beta}\; v^{\alpha}\left(\partial_{\alpha} g_{\mu\beta} - \partial_{\beta} g_{\mu\alpha} + \partial_{\mu} g_{\alpha\beta}\right) + \partial_{\mu} v^{\nu}$

$= \frac{1}{2}\; g^{\nu\beta}\; v^{\alpha}\; \partial_{\alpha} g_{\mu\beta} - \frac{1}{2}\; g^{\nu\beta}\; v^{\alpha}\; \partial_{\beta} g_{\mu\alpha} + \frac{1}{2}\; g^{\nu\beta}\; v^{\alpha}\; \partial_{\mu} g_{\alpha\beta} + \partial_{\mu} v^{\nu}$

In[ ]:=
```
Print[
 CD[-μ]@v[ν]                                                    ,
 "= ",
 CD[-μ]@v[ν]  // CovDToChristoffel //
  ScreenDollarIndices                              , "\n = ",
 CD[-μ]@v[ν]  // CovDToChristoffel // ChristoffelToGradMetric //
  ScreenDollarIndices, "\n = ",
 CD[-μ]@v[ν]  // CovDToChristoffel // ChristoffelToGradMetric // ToCanonical //
  ScreenDollarIndices, "\n"
]
```

$$\nabla_\mu v^\nu = \Gamma[\nabla]^\nu{}_{\mu\alpha} v^\alpha + \partial_\mu v^\nu$$

$$= \frac{1}{2} g^{\nu\beta} v^\alpha \left(\partial_\alpha g_{\mu\beta} - \partial_\beta g_{\mu\alpha} + \partial_\mu g_{\alpha\beta}\right) + \partial_\mu v^\nu$$

$$= \frac{1}{2} g^{\nu\beta} v^\alpha \partial_\alpha g_{\mu\beta} - \frac{1}{2} g^{\nu\beta} v^\alpha \partial_\beta g_{\mu\alpha} + \frac{1}{2} g^{\nu\beta} v^\alpha \partial_\mu g_{\alpha\beta} + \partial_\mu v^\nu$$

In[ ]:=
```
Print[
 CD[-ρ]@CD[-μ]@
   v[ν]                                                      ,    "=
   ",
 CD[-ρ]@CD[-μ]@v[ν]  // CovDToChristoffel //
  ScreenDollarIndices                              , "\n = ",
 CD[-ρ]@CD[-μ]@v[ν]  // CovDToChristoffel // ChristoffelToGradMetric //
  ScreenDollarIndices, "\n = ",
 CD[-ρ]@CD[-μ]@v[ν]  // CovDToChristoffel // ChristoffelToGradMetric // ToCanonical //
  ScreenDollarIndices, "\n"
]
```

ToCanonical: Detected metric–incompatible derivatives {PD}.

ToCanonical: Detected metric–incompatible derivatives {PD}.

ToCanonical: Detected metric–incompatible derivatives {PD}.

General: Further output of ToCanonical::cmods will be suppressed during this calculation.

$$\triangledown_\rho \triangledown_\mu v^\nu = -\Gamma[\triangledown]^\beta{}_{\rho\mu} \left(\Gamma[\triangledown]^\nu{}_{\beta\alpha}\ v^\alpha + \partial_\beta v^\nu\right) + \Gamma[\triangledown]^\nu{}_{\rho\beta} \left(\Gamma[\triangledown]^\beta{}_{\mu\alpha}\ v^\alpha + \partial_\mu v^\beta\right) + v^\alpha\ \partial_\rho\Gamma[\triangledown]^\nu{}_{\mu\alpha} + \Gamma[\triangledown]^\nu{}_{\mu\alpha}\ \partial_\rho v^\alpha + \partial_\rho\partial_\mu v^\nu$$

$$= \frac{1}{2}\ g^{\nu\gamma} \left(\frac{1}{2}\ g^{\beta\delta}\ v^\alpha \left(\partial_\alpha g_{\mu\delta} - \partial_\delta g_{\mu\alpha} + \partial_\mu g_{\alpha\delta}\right) + \partial_\mu v^\beta\right)\left(\partial_\beta g_{\rho\gamma} - \partial_\gamma g_{\rho\beta} + \partial_\rho g_{\beta\gamma}\right) -$$
$$\frac{1}{2}\ g^{\beta\gamma} \left(\partial_\beta v^\nu + \frac{1}{2}\ g^{\nu\delta}\ v^\alpha \left(\partial_\alpha g_{\beta\delta} + \partial_\beta g_{\alpha\delta} - \partial_\delta g_{\beta\alpha}\right)\right)\left(-\partial_\gamma g_{\rho\mu} + \partial_\mu g_{\rho\gamma} + \partial_\rho g_{\mu\gamma}\right) +$$
$$\frac{1}{2}\ g^{\nu\beta} \left(\partial_\alpha g_{\mu\beta} - \partial_\beta g_{\mu\alpha} + \partial_\mu g_{\alpha\beta}\right)\partial_\rho v^\alpha +$$
$$\frac{1}{2}\ v^\alpha \left(-\ g^{\beta\delta}\ g^{\nu\gamma} \left(\partial_\alpha g_{\mu\beta} - \partial_\beta g_{\mu\alpha} + \partial_\mu g_{\alpha\beta}\right)\partial_\rho g_{\gamma\delta} +\ g^{\nu\beta} \left(\partial_\rho\partial_\alpha g_{\mu\beta} - \partial_\rho\partial_\beta g_{\mu\alpha} + \partial_\rho\partial_\mu g_{\alpha\beta}\right)\right) + \partial_\rho\partial_\mu v^\nu$$

$$= \frac{1}{2}\ g^{\nu\beta}\ v^\alpha\ \partial_\alpha\partial_\rho g_{\mu\beta} - \Gamma[\triangledown]_{\alpha\mu\rho}\ \partial^\alpha v^\nu - \frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\alpha g_{\mu\gamma}\,\partial_\beta g_{\rho\delta} -$$
$$\frac{1}{2}\ g^{\nu\beta}\ v^\alpha\ \partial_\beta\partial_\rho g_{\mu\alpha} + \frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\beta g_{\rho\delta}\,\partial_\gamma g_{\mu\alpha} + \frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\alpha g_{\beta\delta}\,\partial_\gamma g_{\mu\rho} -$$
$$\frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\beta g_{\alpha\delta}\,\partial_\gamma g_{\mu\rho} + \frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\gamma g_{\mu\rho}\,\partial_\delta g_{\alpha\beta} + \frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\alpha g_{\mu\gamma}\,\partial_\delta g_{\rho\beta} -$$
$$\frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\gamma g_{\mu\alpha}\,\partial_\delta g_{\rho\beta} - \frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\beta g_{\rho\gamma}\,\partial_\mu g_{\alpha\delta} + \frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\gamma g_{\rho\beta}\,\partial_\mu g_{\alpha\delta} -$$
$$\frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\alpha g_{\beta\delta}\,\partial_\mu g_{\rho\gamma} + \frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\beta g_{\alpha\delta}\,\partial_\mu g_{\rho\gamma} - \frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\delta g_{\alpha\beta}\,\partial_\mu g_{\rho\gamma} +$$
$$\frac{1}{2}\ g^{\nu\beta}\ \partial_\alpha g_{\rho\beta}\,\partial_\mu v^\alpha - \frac{1}{2}\ g^{\nu\beta}\ \partial_\beta g_{\rho\alpha}\,\partial_\mu v^\alpha + \frac{1}{2}\ g^{\nu\beta}\ \partial_\mu v^\alpha\,\partial_\rho g_{\alpha\beta} - \frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\alpha g_{\mu\gamma}\,\partial_\rho g_{\beta\delta} +$$
$$\frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\gamma g_{\mu\alpha}\,\partial_\rho g_{\beta\delta} - \frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\mu g_{\alpha\gamma}\,\partial_\rho g_{\beta\delta} - \frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\alpha g_{\beta\delta}\,\partial_\rho g_{\mu\gamma} +$$
$$\frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\beta g_{\alpha\delta}\,\partial_\rho g_{\mu\gamma} - \frac{1}{4}\ g^{\gamma\delta}\ g^{\nu\beta}\ v^\alpha\ \partial_\delta g_{\alpha\beta}\,\partial_\rho g_{\mu\gamma} + \frac{1}{2}\ g^{\nu\beta}\ \partial_\alpha g_{\mu\beta}\,\partial_\rho v^\alpha -$$
$$\frac{1}{2}\ g^{\nu\beta}\ \partial_\beta g_{\mu\alpha}\,\partial_\rho v^\alpha + \frac{1}{2}\ g^{\nu\beta}\ \partial_\mu g_{\alpha\beta}\,\partial_\rho v^\alpha + \frac{1}{2}\ g^{\nu\beta}\ v^\alpha\ \partial_\rho\partial_\mu g_{\alpha\beta} + \partial_\rho\partial_\mu v^\nu$$

$\nabla_\rho\left(u^\mu \nabla_\mu S^{\rho\nu}\right)$

In[ ]:=
```
Print[
 CD[-ρ]@(u[μ] CD[-μ]@S[ρ, ν]), " = ",
 CD[-ρ]@(u[μ] CD[-μ]@S[ρ, ν]) // CovDToChristoffel // ScreenDollarIndices
]
```

$$\left(\triangledown_\mu S^{\rho\nu}\right)\left(\triangledown_\rho u^\mu\right) + u^\mu \left(\triangledown_\rho \triangledown_\mu S^{\rho\nu}\right) = \left(\Gamma[\triangledown]^\rho{}_{\mu\beta}\ S^{\beta\nu} + \Gamma[\triangledown]^\nu{}_{\mu\alpha}\ S^{\rho\alpha} + \partial_\mu S^{\rho\nu}\right)\left(\Gamma[\triangledown]^\mu{}_{\rho\gamma}\ u^\gamma + \partial_\rho u^\mu\right) +$$
$$u^\mu \left(-\Gamma[\triangledown]^\gamma{}_{\rho\mu} \left(\Gamma[\triangledown]^\rho{}_{\gamma\beta}\ S^{\beta\nu} + \Gamma[\triangledown]^\nu{}_{\gamma\alpha}\ S^{\rho\alpha} + \partial_\gamma S^{\rho\nu}\right) +\right.$$
$$\Gamma[\triangledown]^\rho{}_{\rho\lambda} \left(\Gamma[\triangledown]^\lambda{}_{\mu\beta}\ S^{\beta\nu} + \Gamma[\triangledown]^\nu{}_{\mu\alpha}\ S^{\lambda\alpha} + \partial_\mu S^{\lambda\nu}\right) + \Gamma[\triangledown]^\nu{}_{\rho\delta} \left(\Gamma[\triangledown]^\rho{}_{\mu\beta}\ S^{\beta\delta} + \Gamma[\triangledown]^\delta{}_{\mu\alpha}\ S^{\rho\alpha} + \partial_\mu S^{\rho\delta}\right) +$$
$$\left. S^{\rho\alpha}\ \partial_\rho\Gamma[\triangledown]^\nu{}_{\mu\alpha} + S^{\beta\nu}\ \partial_\rho\Gamma[\triangledown]^\rho{}_{\mu\beta} + \Gamma[\triangledown]^\rho{}_{\mu\beta}\ \partial_\rho S^{\beta\nu} + \Gamma[\triangledown]^\nu{}_{\mu\alpha}\ \partial_\rho S^{\rho\alpha} + \partial_\rho\partial_\mu S^{\rho\nu}\right)$$

Lie Derivatives

In[ ]:=
```
Print[
 LieD[v[μ]][u[ν]], " = ",
 LieD[v[μ]][u[ν]] // LieDToCovD // ScreenDollarIndices, " = ",
 LieDToCovD[LieD[v[μ]][u[ν]], CD] // ScreenDollarIndices
]
```

$\mathcal{L}_v u^\nu = v^\alpha \partial_\alpha u^\nu - u^\alpha \partial_\alpha v^\nu = v^\alpha (\nabla_\alpha u^\nu) - u^\alpha (\nabla_\alpha v^\nu)$

Lie Brackets:

In[ ]:=
```
Print[
 Bracket[u[μ], v[ν]][σ], " = ",
 Bracket[u[μ], v[ν]][σ] // BracketToCovD // ScreenDollarIndices, " = ",
 BracketToCovD[Bracket[u[μ], v[ν]][σ], CD] // ScreenDollarIndices
]
```

$[u^\mu, v^\nu]^\sigma = - v^\alpha \partial_\alpha u^\sigma + u^\alpha \partial_\alpha v^\sigma = - v^\alpha (\nabla_\alpha u^\sigma) + u^\alpha (\nabla_\alpha v^\sigma)$

Notice the Leibniz rule and the linearity in the 1st slot:

In[ ]:=
```
Print[
 LieD[v[μ]][F[-ν, -ρ]], " = ",
 LieD[v[μ]][F[-ν, -ρ]] // LieDToCovD // ScreenDollarIndices, " = ",
 LieDToCovD[LieD[v[μ]][F[-ν, -ρ]], CD] // ScreenDollarIndices, "\n",
 LieD[v[μ]][F[-ν, -ρ] u[σ]], "\n",
 LieD[v[μ] + 7 GNewton u[μ]][F[-ν, -ρ]], "\n",
 LieD[ u[μ]][GNewton v[ρ]], "\n",
 LieD[ u[μ]][ f[]      v[ρ]]
]
```

$\mathcal{L}_v F_{\nu\rho} = v^\alpha \partial_\alpha F_{\nu\rho} + F_{\alpha\rho} \partial_\nu v^\alpha + F_{\nu\alpha} \partial_\rho v^\alpha = v^\alpha (\nabla_\alpha F_{\nu\rho}) + F_{\alpha\rho} (\nabla_\nu v^\alpha) + F_{\nu\alpha} (\nabla_\rho v^\alpha)$

$u^\sigma (\mathcal{L}_v F_{\nu\rho}) + F_{\nu\rho} (\mathcal{L}_v u^\sigma)$

$7 G_N (\mathcal{L}_u F_{\nu\rho}) + \mathcal{L}_v F_{\nu\rho}$

$G_N (\mathcal{L}_u v^\rho)$

$v^\rho (\mathcal{L}_u f) + f (\mathcal{L}_u v^\rho)$

In[ ]:=
```
Bracket[v[μ] + GNewton u[λ], ω[ρ]][σ]
```

Out[ ]= $G_N [u^\lambda, \omega^\rho]^\sigma + [v^\mu, \omega^\rho]^\sigma$

In[ ]:=
```
Bracket[ f[] u[λ],  v[ρ]][σ]
```
Out[ ]=
$f\,[u^{\lambda}, v^{\rho}]^{\sigma} - u^{\sigma}\,\partial_{v} f$

Prove that $\mathcal{L}_{[u,v]}\,\xi = [\mathcal{L}_u, \mathcal{L}_v]\,\xi$

In[ ]:=
```
Print[
 "-----------------------------------------------------------------------------------
   ----------------\n",
 "The left hand side:  ℒ[u,v] ξ\n",
                 LieD[              Bracket[ u[μ], v[ν]][σ]      ][
  ξ[ρ]]                              , " = ",
                 LieD[ BracketToCovD[Bracket[ u[μ], v[ν]][σ], CD ]][ ξ[ρ] ]      //
  ScreenDollarIndices, " = ",
 eq1 = LieDToCovD[LieD[ BracketToCovD[Bracket[ u[μ], v[ν]][σ], CD ]][ ξ[ρ] ], CD] //
   ScreenDollarIndices, "\n",
 "-----------------------------------------------------------------------------------
   ----------------\n",
 "The 1st term of the RHS: ℒuℒvξ \n",
 LieD [u[μ]] [ LieD [v[ν]] [ξ[ρ]] ] , " = ",
 eq2 = LieDToCovD[LieD [u[μ]] [ LieD [v[ν]] [ξ[ρ]] ], CD] // ScreenDollarIndices, "\n",
 "-----------------------------------------------------------------------------------
   ----------------\n",
 "The 2nd term of the RHS: ℒvℒuξ \n",
 LieD [v[ν]] [ LieD [u[μ]] [ξ[ρ]] ] , " = ",
 eq3 = LieDToCovD[LieD [v[ν]] [ LieD [u[μ]] [ξ[ρ]] ], CD] // ScreenDollarIndices, "\n",
 "-----------------------------------------------------------------------------------
   ----------------\n",
 "The difference: ℒ[u,v]ξ - [ℒu ,ℒv]ξ =ℒ[u,v]ξ - ℒuℒvξ + ℒvℒuξ  \n\n",
 LieD[              Bracket[ u[μ], v[ν]][σ]      ][ ξ[ρ] ] -
  LieD [u[μ]] [ LieD [v[ν]] [ξ[ρ]] ] + LieD [v[ν]] [ LieD [u[μ]] [ξ[ρ]] ] , " = \n",
 eq4 = (eq1 - eq2 + eq3 // Simplification), " = \n",
 eq5 = (CommuteCovDs[eq4, CD, {-α, -β}] // ScreenDollarIndices), " = \n",
 eq6 = (eq5 // Simplification), " = \n",
 eq7 = eq6 /. {RiemannCD[ρ, -α, -β, -γ] →
     -(RiemannCD[ρ, -γ, -α, -β] + RiemannCD[ρ, -β, -γ, -α])}, " = \n",
 eq7 // ToCanonical, "\n",
 "-----------------------------------------------------------------------------------
   ----------------\n"
]
```

--
The left hand side: $\mathcal{L}_{[u,v]}\,\xi$

$$\mathcal{L}_{[u^\mu,v^\nu]}\,\xi^\rho = -\left(\mathcal{L}_{v^\alpha\,(\nabla_\alpha u^\sigma)}\,\xi^\rho\right)+\mathcal{L}_{u^\alpha\,(\nabla_\alpha v^\sigma)}\,\xi^\rho = -\,\xi^\beta\left(\nabla_\alpha v^\rho\right)\left(\nabla_\beta u^\alpha\right)+\xi^\beta\left(\nabla_\alpha u^\rho\right)\left(\nabla_\beta v^\alpha\right)- v^\alpha\left(\nabla_\alpha u^\beta\right)\left(\nabla_\beta \xi^\rho\right)+u^\alpha\left(\nabla_\alpha v^\beta\right)\left(\nabla_\beta \xi^\rho\right)+v^\alpha\,\xi^\beta\left(\nabla_\beta\nabla_\alpha u^\rho\right)-u^\alpha\,\xi^\beta\left(\nabla_\beta\nabla_\alpha v^\rho\right)$$

--
The 1st term of the RHS: $\mathcal{L}_u\mathcal{L}_v\xi$

$$\mathcal{L}_u\,\mathcal{L}_v\,\xi^\rho = \xi^\alpha\left(\nabla_\alpha v^\beta\right)\left(\nabla_\beta u^\rho\right)-v^\alpha\left(\nabla_\alpha \xi^\beta\right)\left(\nabla_\beta u^\rho\right)+ u^\beta\left(\nabla_\alpha \xi^\rho\right)\left(\nabla_\beta v^\alpha\right)-u^\beta\left(\nabla_\alpha v^\rho\right)\left(\nabla_\beta \xi^\alpha\right)-u^\beta\,\xi^\alpha\left(\nabla_\beta\nabla_\alpha v^\rho\right)+u^\beta\,v^\alpha\left(\nabla_\beta\nabla_\alpha \xi^\rho\right)$$

--
The 2nd term of the RHS: $\mathcal{L}_v\mathcal{L}_u\xi$

$$\mathcal{L}_v\,\mathcal{L}_u\,\xi^\rho = v^\beta\left(\nabla_\alpha \xi^\rho\right)\left(\nabla_\beta u^\alpha\right)+\xi^\alpha\left(\nabla_\alpha u^\beta\right)\left(\nabla_\beta v^\rho\right)- u^\alpha\left(\nabla_\alpha \xi^\beta\right)\left(\nabla_\beta v^\rho\right)-v^\beta\left(\nabla_\alpha u^\rho\right)\left(\nabla_\beta \xi^\alpha\right)-v^\beta\,\xi^\alpha\left(\nabla_\beta\nabla_\alpha u^\rho\right)+u^\alpha\,v^\beta\left(\nabla_\beta\nabla_\alpha \xi^\rho\right)$$

--
The difference: $\mathcal{L}_{[u,v]}\xi - [\mathcal{L}_u,\mathcal{L}_v]\xi = \mathcal{L}_{[u,v]}\xi - \mathcal{L}_u\mathcal{L}_v\xi + \mathcal{L}_v\mathcal{L}_u\xi$

$$-\left(\mathcal{L}_u\,\mathcal{L}_v\,\xi^\rho\right)+\mathcal{L}_v\,\mathcal{L}_u\,\xi^\rho+\mathcal{L}_{[u^\mu,v^\nu]}\,\xi^\rho =$$

$$v^\alpha\,\xi^\beta\left(-\left(\nabla_\alpha\nabla_\beta u^\rho\right)+\nabla_\beta\nabla_\alpha u^\rho\right)+u^\alpha\left(\xi^\beta\left(\nabla_\alpha\nabla_\beta v^\rho-\nabla_\beta\nabla_\alpha v^\rho\right)+v^\beta\left(-\left(\nabla_\alpha\nabla_\beta \xi^\rho\right)+\nabla_\beta\nabla_\alpha \xi^\rho\right)\right) =$$

$$R[\nabla]_{\alpha\beta\gamma}{}^{\rho}\,u^\gamma\,v^\alpha\,\xi^\beta+u^\alpha\left(-R[\nabla]_{\alpha\beta\gamma}{}^{\rho}\,v^\gamma\,\xi^\beta+R[\nabla]_{\alpha\beta\delta}{}^{\rho}\,v^\beta\,\xi^\delta\right) =$$

$$-\left(\left(R[\nabla]^\rho{}_{\alpha\beta\gamma}-R[\nabla]^\rho{}_{\beta\alpha\gamma}+R[\nabla]^\rho{}_{\gamma\alpha\beta}\right)u^\alpha\,v^\beta\,\xi^\gamma\right) =$$

$$-\left(\left(-R[\nabla]^\rho{}_{\beta\alpha\gamma}-R[\nabla]^\rho{}_{\beta\gamma\alpha}\right)u^\alpha\,v^\beta\,\xi^\gamma\right) =$$

$$0$$

--

Acknowledgements

This notebook has been programmed by Konstantinos Anagnostopoulos, Physics Department, National Technical University of Athens, Greece, while he was an instructor of the 4th year undergraduate course "General Relativity and Cosmology". It was created for fun, but it may turn out to be useful to everyone studying the General Theory of Relativity for the first time.
Email: konstant@mail.ntua.gr
Web: http://physics.ntua.gr/konstant

programming part or the text part of the notebook).